#-------------------------------------------------
#
# Project created by QtCreator 2015-07-11T22:07:34
#
#-------------------------------------------------

QT       += gui core

TARGET = report
TEMPLATE = lib

DEFINES += REPORT_LIBRARY

SOURCES += \
    report.cpp \
    textreport.cpp

HEADERS += \
    report_global.h \
    report.h \
    textreport.h \
    report_p.h \
    textreport_p.h

include($$PWD/../common.pri)



defineTest(copyToDestdir) {
    files = $$1

    for(FILE, files) {
        DDIR = $$DESTDIR
        # Replace slashes in paths with backslashes for Windows
        win32:FILE ~= s,/,\\,g
        win32:DDIR ~= s,/,\\,g
        QMAKE_POST_LINK += $$QMAKE_COPY $$quote($$FILE) $$quote($$DDIR) $$escape_expand(\\n\\t)
    }
    export(QMAKE_POST_LINK)
}

TESTFILES += \
    $$PWD/report.h \
    $$PWD/textreport.h \
    $$PWD/report_global.h


copyToDestdir($$TESTFILES)

unix {
    target.path = /usr/lib
    INSTALLS += target
}
